# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Arch Haskell Team <arch-haskell@haskell.org>

_hkgname=regex-compat-tdfa
pkgname=haskell-regex-compat-tdfa
pkgver=0.95.1.4
pkgrel=186
pkgdesc="One module layer over regex-tdfa to replace Text.Regex."
url="https://hub.darcs.net/shelarcy/regex-compat-tdfa"
license=("BSD")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-regex-base' 'haskell-regex-tdfa')
makedepends=('ghc')
source=("https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz")
sha512sums=('7e387a7713ff8e43cfce4262b1288b85f8b60729162a842186bf6a7e1078dfb9e26013087ff7e9bcf81521f8c352926873bb280b9df57b016aef2ca4ce70aa14')

prepare() {
    cd $_hkgname-$pkgver
    sed -i 's/!0/! 0/' Text/Regex.hs
}

build() {
    cd $_hkgname-$pkgver
    
    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

package() {
    cd $_hkgname-$pkgver
    
    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
